```
> restart;
> # Boundary condtions at the leading order along line y=0
  #
  P1x := w·cos(k·x − w·t);
  P1y := 0;
```

$$P1x := w\cos(k\,x - t\,w)$$

$$P1y := 0 \tag{1}$$

```
> # Form of the solution
  #
  psi1 := (x, y) → (A1·exp(-k·y) + B1·y·exp(-k·y) )·sin(k·x − t·w);
```

$$\psi 1 := (x, y) \to \left(A1\, e^{-k\,y} + B1\, y\, e^{-k\,y}\right) \sin(k\,x - t\,w) \tag{2}$$

```
> # Apply boundary condtions to find constants
  #
  solve({D[1](psi1)(x, 0) = P1x, D[2](psi1)(x, 0) = P1y}, {A1, B1});
```

$$\left\{A1 = \frac{w}{k},\ B1 = w\right\} \tag{3}$$

```
> assign(%);
> # Leading order O(eps) solution for the streamfunction
  #
  psi1(x, y);
```

$$\left(\frac{w\, e^{-k\,y}}{k} + w\, y\, e^{-k\,y}\right) \sin(k\,x - t\,w) \tag{4}$$

```
> # Boundary condtions for the second-order solution
  #
  P2x := -D[1, 2](psi1)(x, 0) · sin(k·x − w·t);
  P2y := -D[2, 2](psi1)(x, 0)·sin(k·x − w·t);
```

$$P2x := 0$$

$$P2y := k\, w \sin(k\,x - t\,w)^2 \tag{5}$$

```
> # express sin^2 as sum of eigenfunctions
  #
  P2y := subs( sin(k·x − w·t)^2 = 1/2 (1 − cos(2·(k·x − w·t))), P2y);
```

$$P2y := k\, w \left(\frac{1}{2} - \frac{1}{2}\cos(2\,k\,x - 2\,t\,w)\right) \tag{6}$$

```
> # Form of the streamfunction at second order
  #
  psi2 := (x, y) → (A2·exp(-k·y) + B2·y·exp(-k·y))·cos(2·(k·x − w·t)) + C2·y;
```

$$\psi 2 := (x, y) \to \left(A2\, e^{-k\,y} + B2\, y\, e^{-k\,y}\right) \cos(2\,k\,x - 2\,t\,w) + C2\, y \tag{7}$$

```
> # Equations for boundary condtions to find constants
  #
  BC1 := D[1](psi2)(x, 0) − P2x;
  BC2 := collect(D[2](psi2)(x, 0) − P2y, cos);
```

$$BC1 := -2\, A2 \sin(2\,k\,x - 2\,t\,w)\, k$$

$$BC2 := \left(-A2\,k + B2 + \frac{1}{2}\,k\,w\right)\cos(2\,k\,x - 2\,t\,w) + C2 - \frac{1}{2}\,k\,w \tag{8}$$

```
> # solve for constants by applying boundary conditions
  #
  A2 := 0;
   C2 := solve(subs(cos(2·k·x − 2· t·w) = 0, BC2), C2);
   B2 := solve(coeff(BC2, cos(2·k·x − 2·t·w)), B2);
  ;
```

$$A2 := 0$$

$$C2 := \frac{1}{2}\,k\,w$$

$$B2 := -\frac{1}{2}\,k\,w \tag{9}$$

```
> # Form of the second-order solution
   #
  simplify(psi2(x, y));
```

$$-\frac{1}{2}\,k\,w\,y\left(e^{-k\,y}\cos(2\,k\,x - 2\,t\,w) - 1\right) \tag{10}$$

```
> # Expression for the swimming speed (times eps^2)
  #
  subs(exp(-k·y) = 0, diff(psi2(x, y), y));
```

$$\frac{1}{2}\,k\,w \tag{11}$$

```
>
```